Problem:
Score : 1100 points

Problem Statement

Let us consider a grid of squares with 10^9 rows and N columns. Let (i, j) be the square at the i-th column (1 \leq i \leq N) from the left and j-th row (1 \leq j \leq 10^9) from the bottom.

Snuke has cut out some part of the grid so that, for each i = 1, 2, ..., N, the bottom-most h_i squares are remaining in the i-th column from the left.
Now, he will paint the remaining squares in red and blue.
Find the number of the ways to paint the squares so that the following condition is satisfied:

Every remaining square is painted either red or blue.

For all 1 \leq i \leq N-1 and 1 \leq j \leq min(h_i, h_{i+1})-1, there are exactly two squares painted red and two squares painted blue among the following four squares: (i, j), (i, j+1), (i+1, j) and (i+1, j+1).

Since the number of ways can be extremely large, print the count modulo 10^9+7.

Constraints

1 \leq N \leq 100

1 \leq h_i \leq 10^9

Input

Input is given from Standard Input in the following format:

N
h_1 h_2 ... h_N

Output

Print the number of the ways to paint the squares, modulo 10^9+7.

Sample Input 1

9
2 3 5 4 1 2 4 2 1

Sample Output 1

12800

One of the ways to paint the squares is shown below:

#
## #
### #
#### ###
#########

Sample Input 2

2
2 2

Sample Output 2

6

There are six ways to paint the squares, as follows:

## ## ## ## ## ##
## ## ## ## ## ##

Sample Input 3

5
2 1 2 1 2

Sample Output 3

256

Every way to paint the squares satisfies the condition.

Sample Input 4

9
27 18 28 18 28 45 90 45 23

Sample Output 4

844733013

Remember to print the number of ways modulo 10^9 + 7.

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program histogram_coloring
  implicit none
  type pair
    integer(8) :: x, y
  end type
  integer(8), parameter :: md = 1000000007_8
  integer :: n, h(100) = 0
  type(pair) :: p
  read(*,*) n
  read(*,*) h(1:n)
  p = solve(h(1:n))
  write(*,'(i0)') p%y
contains
  recursive function solve(h) result(ret)
    integer, intent(in) :: h(:)
    integer :: m, x(size(h)+1)
    integer :: n, i, j, w
    integer(8) :: dp1, dp2
    type(pair) :: ret, tmp
    n = size(h)
    m = minval(h)
    x(1:n) = h-m
    x(n+1) = 0
    w = n
    dp1 = 1_8
    dp2 = 1_8
    i = 0
    do while (i <= n)
      j = i+1
      do while (x(j) /= 0)
        j = j+1
      end do
      if (i+1 <= j-1) then
        tmp = solve(x(i+1:j-1))
        dp1 = mod(dp1*tmp%x,md)
        dp2 = mod(dp2*(tmp%x+tmp%y),md)
        w = w-(j-i-1)
      end if
      i = j
    end do
    ret%x = mod(dp1*pow(2_8,m),md)
    ret%y = modulo(mod(dp2*pow(2_8,w),md)+mod(dp1*(pow(2_8,m)-2_8),md),md)
  end
  function pow(a,b) result(r)
    integer(8), intent(in) :: a
    integer, intent(in) :: b
    integer(8) :: r, p
    integer :: k
    r = 1_8
    p = a
    k = b
    do while (k > 0)
      if (btest(k,0)) r = mod(r*p,md)
      p = mod(p*p,md)
      k = rshift(k,1)
    end do
  end
end program histogram_coloring